Alkuehtoa vastaavan yksittäisratkaisun etsiminen

Muodostamalla ensin yleinen ratkaisu

Olkoon annettuna differentiaaliyhtälö alkuehtoineen:

```
diffyht = x^2 y''[x] - 3 x y'[x] + 4 y[x] == x^3
```

$4\,y[x] - 3\,x\,y'[x] + x^2\,y''[x] == x^3$

```
ehto1 = y[2] == 3
```

y[2] == 3

```
ehto2 = y'[2] == 4
```

y′[2] == 4

Yleinen ratkaisu funktion muodossa saadaan `DSolve`-funktiolla:

```
ratkaisu = DSolve[diffyht, y, x]
```

$\{\{y \to \text{Function}[\{x\}, x^3 + x^2\,C[1] + 2\,x^2\,C[2]\,\text{Log}[x]]\}\}$

Sijoittamalla tämä alkuehtoihin saadaan algebralliset yhtälöt vakioiden `C[1]` ja `C[2]` määrittämiseksi:

```
yht1 = ehto1 /. First[ratkaisu]
```

8 + 4 C[1] + 8 C[2] Log[2] == 3

```
yht2 = ehto2 /. First[ratkaisu]
```

12 + 4 C[1] + 4 C[2] + 8 C[2] Log[2] == 4

Näistä voidaan ratkaista vakioiden arvot:

```
vakiot = Solve[{yht1, yht2}, {C[1], C[2]}]
```

$\left\{\left\{C[1] \to -\frac{1}{4}\,(5 - 6\,\text{Log}[2]),\ C[2] \to -\frac{3}{4}\right\}\right\}$

Sijoittamalla vakiot yleiseen ratkaisuun saadaan alkuehtoa vastaava yksittäisratkaisu:

```
y0 = y /. First[ratkaisu] /. First[vakiot]
```

$$\text{Function}\left[\{x\},\ x^3 + \frac{1}{4}\,x^2\,(-(5 - 6\,\text{Log}[2])) + \frac{2}{4}\,x^2\,(-3)\,\text{Log}[x]\right]$$

Sievennettynä ratkaisun lauseke on

```
y0[x] // Simplify
```

$$\frac{1}{4}\,x^2\,(-5 + 4\,x + \text{Log}[64] - 6\,\text{Log}[x])$$

Tämä toteuttaa differentiaaliyhtälön ja alkuehdot:

```
{diffyht, ehto1, ehto2} /. y -> y0 // Simplify
```

{True, True, True}

Ratkaisemalla suoraan `DSolve`-komennolla

Alkuehdot voidaan yhtälön ohella antaa `DSolve`-komennolle, jolloin yksittäisratkaisu saadaan suoraan:

```
ratkaisu = DSolve[{diffyht, ehto1, ehto2}, y, x]
```

$$\left\{\left\{y \to \text{Function}\left[\{x\},\ \frac{1}{4}\,(-5\,x^2 + 4\,x^3 + 6\,x^2\,\text{Log}[2] - 6\,x^2\,\text{Log}[x])\right]\right\}\right\}$$

Ratkaisufunktio ja sen lauseke ovat tällöin

```
y0 = y /. First[ratkaisu]
```

$$\text{Function}\left[\{x\},\ \frac{1}{4}\,(-5\,x^2 + 4\,x^3 + 6\,x^2\,\text{Log}[2] - 6\,x^2\,\text{Log}[x])\right]$$

```
y0[x] // Simplify
```

$$\frac{1}{4}\,x^2\,(-5 + 4\,x + \text{Log}[64] - 6\,\text{Log}[x])$$

Linkit

yleinen ja yksittäisratkaisu
alkuehto
ratkaiseminen algebrallisesti Mathematicalla (symalg.nb)
ratkaisun sijoittaminen yhtälöön Mathematican keinoin (rtksij.nb)

SKK 27.04.2001